
Queries to find malware
Orphaned Processes
Determine if a process is orphaned
Analyst notes
To be precise, this query looks for a parent process record that exists in the same timeframe as
the child process record.

Known false positives
Per notes above, if the parent process was created outside of the search timeframe, the
process will appear to be orphaned.

Known false negatives
None.
aid=<aid> <process_id> event_simpleName=ProcessRollup2
| eval JoinId=ParentProcessId_decimal
| rename CommandLine as ChildCommandLine
| join type=outer JoinId
[search aid=<aid> event_simpleName=ProcessRollup2
| eval JoinId=TargetProcessId_decimal
| rename CommandLine as ParentCommandLine]
| eval ParentCommandLine=coalesce(ParentCommandLine,"IamAnOrphan")

Find orphaned processes for a host v1 (filtering on common hashes)
Analyst notes
Timeframe is important for this query. Splunk’s default configuration only allows 10,000 records
to be returned for each subsearch.
I didn’t experiment with this, but it’s possible that the map command could be used to run this
multiple hosts.
I had to do a significant amount of string-based filtering to get the false positives down to a
manageable level. A better implementation would be to have a full and complete lookup table
showing how common/rare each SHA256 hash is. This option was unavailable to me because
we are limited by what CrowdStrike allows us to on their Splunk cluster.
This is a long query so here is a summary:
- Search for all processes for a given agent, excluding some common system process
names.
- Rename the CommandLine argument of all the processes found to the name
ChildCommandLine and create a new variable to hold the process id we will join on.
- Calculate the duration of all the processes based on the EndOfProcess record, if the
record exists, or the current time.
- Search for all processes for this agent over the timeframe to find the parents and add a
ParentCommandLine to the child process record, if the parent process record is found.
- Exclude all processes where a parent was found.
- Create a summary table which calculates the number of agents each SHA256 was found
on, the maximum duration for any records with that hash, and the commands for each of
those hashes.

-

-

Exclude all processes found on more than one host (yes, this is redundant if we’re
searching for just one agent, but I left it in for future-proofing)
Search for any SHA256 hashes that have triggered virustotal detections during the
timeframe and add that count to the record, if there was a hit
Find the 10,000 most and least common SHA256 hashes for all Macs over the
timeframe, and calculate how many hosts each hash appeared on. This is a bit of a
hack. There are too many records for me to build a complete lookup table to join on, so I
had to settle for this.
Exclude any records where the hash was found on more than one host in the
environment, based on the 10,000 most and least common SHA256 hash records.

Known false positives
If the parent process was created outside of the search timeframe, the process will appear to be
orphaned. That’s why so many commands are filtered at the beginning - this is an attempt to
exclude long-running system processes whose parents were probably created outside of the
time frame.
If there are more than 10,000 process (or EndOfProcess) records for the host in the timeframe,
the subsearches will return incomplete results and some processes will be marked as orphans
that aren’t. Additionally, the duration of the processes will be artificially inflated because not all
of the EndOfProcess records will be returned.

Known false negatives
If the malicious orphaned process contains a string that is excluded in the first part of the query,
it won’t show up. These are mostly privileged system directories, so if the attacker was able to
write to them then the attacker has already elevated their privileges. Nevertheless,
a better approach would be to have a complete list of of SHA256 hashes known to be common
for the environment. The rest are tools internal to our environment.
event_platform=Mac aid=<aid> event_simpleName=ProcessRollup2 NOT
CommandLine="/System/*" NOT CommandLine="/Library/*" NOT CommandLine="/usr/libexec/*"
NOT CommandLine=xpcproxy* NOT CommandLine="/Applications/Utilities/*" NOT
CommandLine="make*" NOT CommandLine=ipconfig* NOT CommandLine="/Applications/*" NOT
"/Users/*/Library/Application Support/*" NOT CommandLine=<bunch_of_internal_stuff>
| eval JoinId=ParentProcessId_decimal
| rename CommandLine as ChildCommandLine
| join type=outer aid,TargetProcessId_decimal [search event_platform=Mac aid=<aid>
event_simpleName=EndOfProcess | rename _time as EndTime | fields
aid,TargetProcessId_decimal, EndTime]
| eval duration=if(isnull(EndTime),now()-_time,EndTime-_time)

| join type=outer JoinId,aid [search event_platform=Mac aid=<aid>
event_simpleName=ProcessRollup2 | eval JoinId=TargetProcessId_decimal | rename
CommandLine as ParentCommandLine | fields JoinId, ParentCommandLine]
| eval ParentCommandLine=coalesce(ParentCommandLine,"IamAnOrphan")
| search ParentCommandLine="IamAnOrphan"
| eval ChildCommandLine=substr(ChildCommandLine,1,50)
| stats values(ChildCommandLine) as Commands, max(duration) as duration, dc(aid) as
AgentsWithHash by SHA256HashData
| search AgentsWithHash=1
| join type=outer SHA256HashData [search event_platform=Mac event_simpleName=VT |
stats sum(detectionCount) as VTCount by sha256 | rename sha256 as SHA256HashData]
| join type=outer SHA256HashData [search event_platform=Mac
event_simpleName=ProcessRollup2 | top SHA256HashData limit=10000 by aid | stats
dc(aid) as CommonGPopCount by SHA256HashData]
| join type=outer SHA256HashData [search event_platform=Mac
event_simpleName=ProcessRollup2 | rare SHA256HashData limit=10000 by aid
| stats dc(aid) as RareGPopCount by SHA256HashData]
| fillnull value=0 CommonGPopCount
| fillnull value=0 RareGPopCount
| fillnull value=0 VTCount
| search CommonGPopCount<2 RareGPopCount<2

Find orphaned processes for a host v2 (filtering based on common hashes
and duration of process)
Analyst notes
Timeframe is important for this query. Splunk’s default configuration only allows 10,000 records
to be returned for each subsearch.
I didn’t experiment with this, but it’s possible that the map command could be used to run this
multiple hosts.
I had to do a significant amount of string-based filtering to get the false positives down to a
manageable level. A better implementation would be to have a full and complete lookup table
showing how common/rare each SHA256 hash is. This option was unavailable to me because
we are limited by what CrowdStrike allows us to on their Splunk cluster.
The 10,000 most and least common SHA256 hashes for all Macs are used to filter results for
rarity. This is a bit of a hack. There are too many records for me to build a complete lookup table
to join on, so I had to settle for this.
The only differences between this query and the first version (documented above) is an
additional filter that looks processes that have run over 10 seconds or were detected by one
vendor on virustotal.

Known false positives
If the parent process was created outside of the search timeframe, the process will appear to be
orphaned. That’s why so many commands are filtered at the beginning - this is an attempt to
exclude long-running system processes whose parents were probably created outside of the
time frame.
If there are more than 10,000 process (or EndOfProcess) records for the host in the timeframe,
the subsearches will return incomplete results and some processes will be marked as orphans
that aren’t. Additionally, the duration of the processes will be artificially inflated because not all
of the EndOfProcess records will be returned.
Since the duration is actually part of the filtering for this query, an incomplete list of
EndOfProcess records will cause false positives.
Since this query looks for any virustotal hit, false positives on virustotal will cause false positives
for this query. A more sophisticated threshold could calculate a rating based on how trustworthy
each virustotal hit is.

Known false negatives
If the malicious orphaned process contains a string that is excluded in the first part of the query,
it won’t show up. These are mostly privileged system directories, so if the attacker was able to
write to them then the attacker has already elevated their privileges. Nevertheless,
a better approach would be to have a complete list of of SHA256 hashes known to be common
for the environment. The rest of the excluded CommandLine values are tools internal to our
environment.
event_platform=Mac aid=<aid> event_simpleName=ProcessRollup2 NOT
CommandLine="/System/*" NOT CommandLine="/Library/*" NOT CommandLine="/usr/libexec/*"
NOT CommandLine=xpcproxy* NOT CommandLine="/Applications/Utilities/*" NOT
CommandLine="make*" NOT CommandLine=ipconfig* NOT CommandLine="/Applications/*" NOT
"/Users/*/Library/Application Support/*" NOT CommandLine=<bunch_of_internal_stuff>
| eval JoinId=ParentProcessId_decimal
| rename CommandLine as ChildCommandLine
| join type=outer aid,TargetProcessId_decimal
[search event_platform=Mac aid=<aid> event_simpleName=EndOfProcess
| rename _time as EndTime
| fields aid,TargetProcessId_decimal, EndTime]
| eval duration=if(isnull(EndTime),now()-_time,EndTime-_time)
| join type=outer JoinId,aid
[search event_platform=Mac aid=<aid> event_simpleName=ProcessRollup2
| eval JoinId=TargetProcessId_decimal
| rename CommandLine as ParentCommandLine
| fields JoinId, ParentCommandLine]
| eval ParentCommandLine=coalesce(ParentCommandLine,"IamAnOrphan")
| search ParentCommandLine="IamAnOrphan"
| eval ChildCommandLine=substr(ChildCommandLine,1,50)
| stats values(ChildCommandLine) as Commands, max(duration) as duration, dc(aid) as
AgentsWithHash by SHA256HashData
| search AgentsWithHash=1
| join type=outer SHA256HashData
[search event_platform=Mac event_simpleName=VT
| stats sum(detectionCount) as VTCount by sha256
| rename sha256 as SHA256HashData]
| join type=outer SHA256HashData
[search event_platform=Mac event_simpleName=ProcessRollup2
| top SHA256HashData limit=10000 by aid
| stats dc(aid) as CommonGPopCount by SHA256HashData]
| join type=outer SHA256HashData
[search event_platform=Mac event_simpleName=ProcessRollup2

|
|
|
|

| rare SHA256HashData limit=10000 by aid
| stats dc(aid) as RareGPopCount by SHA256HashData]
fillnull value=0 CommonGPopCount
fillnull value=0 RareGPopCount
fillnull value=0 VTCount
search CommonGPopCount<2 RareGPopCount<2 (duration>10 OR VTCount>0)

Malicious Word Documents
Processes typically associated with Word
Analyst notes
The first part of the query (up until the stats command) is a bit inexact because process ids
aren’t globally unique. The tuple (aid, process id) is unique. When the subsearch returns the list
of process ids, the main search may return an unrelated process on a different machine.
However, this should be sorted out by the last filter, which requires at least one of the
executables have the path /Applications/*Microsoft Word*/. I chose to implement it this way
because it’s the simplest way.

Known false positives
If a process tree contains the path /Applications/*Microsoft Word*/ but isn’t related to a user
running Word, this search will pick that up.

Known false negatives
Splunk limits subsearch results to 10,000 by default. This depends on the configuration of
course, but if run over larger time frames this search won’t be complete.
event_platform=Mac event_simpleName=*ProcessRollup2
[search event_simpleName=*ProcessRollup2 event_platform=Mac
CommandLine="/Applications/*Microsoft Word*"
| fields ProcessGroupId_decimal ]
| stats values(CommandLine) as Commands, count by aid,ProcessGroupId_decimal
| search Commands="/Applications/*Microsoft Word*"

Normal:

Abnormal:

Find out what processes Word launched that aren’t Word
Known false negatives
If there are more than 10,000 Word processes for this host, the subsearch will not return all the
records, and the overall results will be incomplete.
aid=<aid> event_simpleName=*ProcessRollup2 NOT CommandLine="/Applications/*Microsoft
Word*" [search aid=<aid> CommandLine="/Applications/*Microsoft Word*"
event_simpleName=*ProcessRollup2
| rename TargetProcessId_decimal as ProcessGroupId_decimal
| return 10000 ProcessGroupId_decimal]
Results from a malicious Word Macro:

Count processes launched by Word
Analyst notes
The dde exploits we’ve seen haven’t necessarily launched oddly named processes, but have
launched more processes than normal. In our environment, we generally see Office products
spawn no more than 2 processes.

Known false negatives
If there are more than 10,000 Word processes for this host, the subsearch will not return all the
records, and the overall results will be incomplete.
aid=<aid> event_simpleName=*ProcessRollup2 NOT CommandLine="/Applications/*Microsoft
Word*" [search aid=<aid> CommandLine="/Applications/*Microsoft Word*"
event_simpleName=*ProcessRollup2
| rename TargetProcessId_decimal as ProcessGroupId_decimal
| return 10000 ProcessGroupId_decimal]
Here’s an example of the behavior of a dde exploit with Excel, which would
show lots of Excel processes spawned if I still had the data to query.

Suspicious Process Behavior
Removing the quarantine attribute on a file
Analyst notes
macOS marks files downloaded with a quarantine attribute. This is used for things like warning a
user when they’re about to run an executable downloaded from the internet. One behavior I
observed in malware is removing this attribute, presumably so users aren’t warned and
automated forensics collection tools (like OSXCollector) don’t flag the file.

Known false positives
I found a few legitimate installers that did this as well. I had to filter what was common in our
environment.

Known false negatives
None.

event_platform=Mac event_simpleName=ProcessRollup2 CommandLine="*xattr -d -r
com.apple.quarantine*" NOT <redacted> NOT <redacted>

Busy process trees
Analyst notes
This query looks for a process tree that contains a large number of processes. The command
line that’s eventually displayed is the non-system process that’s at the top of the tree. For
example, if Word triggered a large number of processes the CommandLine displayed would for
Microsoft Word, not the OS process that launched Word based on user input.

Known false positives
Our inventory and management tools will also appear as busy process trees, thus the filtering.

Known false negatives
Malicious processes that contain the keywords filtered in the last step won’t be shown. A
possible improvement would be to compare against a list of known good hashes, though the
root process might be good (as in Microsoft Word) and still be evidence of malware (in the case
of an infected attachment).
Default Splunk configurations restrict the map command to running 10,000 times. If the first part
of the search below (event_platform=Mac event_simpleName=ProcessRollup2
) returns more
than 10,000 results, only 10,000 of them will be processed by the map command and your
overall results will be incomplete.
event_platform=Mac event_simpleName=ProcessRollup2

| stats count by ProcessGroupId_decimal,aid | search count>50 | map search="search
aid=$aid$ ProcessGroupId_decimal=$ProcessGroupId_decimal$
TargetProcessId_decimal=$ProcessGroupId_decimal$"
| search NOT CommandLine=<redacted> NOT CommandLine=<redacted>

Processes running from tmp dirs
Analyst notes
Known false positives
Very dependent on environment. There will probably be some, so experiment and filter as
needed.

Known false negatives
Malicious processes that contain the keywords filtered won’t be shown. A possible improvement
would be to compare against a list of known good hashes.
event_platform=Mac event_simpleName=ProcessRollup2 (CommandLine="/tmp/*" OR
CommandLine="/private/tmp/*") NOT <redacted> NOT <redacted> NOT <redacted>

Processes running from /Library/Scripts
Analyst notes
In my testing, this was remarkably false positive free.

Known false positives
Dependent on environment.

Known false negatives
None.
event_platform=Mac CommandLine="/Library/Scripts/*"

Copy from tmp directories to user directories
Analyst notes
The few false positives I found in my testing couldn’t easily be filtered out. Because the binary in
this case is cp, you can’t filter on the hash of the executable. A more complex filter would break
apart the command line to determine the source and destination for the copy command, but
going to that effort seemed overkill given the false positive rate I observed.

Known false positives
Dependent on environment.

Known false negatives
None.
event_platform=Mac event_simpleName=ProcessRollup2 FileName=cp
CommandLine="*tmp*Users*"

Long-running process with few network connections - no filtering
Analyst notes
Because we’re using subsearches and subsearches have stricter memory constraints, time
frame is a concern for this query. The default Splunk configuration only allows 10,000 results to
be returned per subsearch.
Inventory management tools may also trigger this, and may need to be filtered out.

Known false positives
Calculating the process duration is based on the results returned by a subsearch. If the number
of EndOfProcess records is more than the limit on subsearches, the results will be incomplete.
That will cause the duration for some of the processes to be incorrectly calculated based on the
current time and artificially inflate the process duration.
Calculating the number of network connections is also based on the results of a subsearch. If
that subsearch returns incomplete results, the NetworkConnectionCount may be lower than
reality and contribute to a false positive.
If a process is terminated uncleanly, the EndOfProcess record may not be present. In these
cases, the process duration will be artificially inflated.

Known false negatives
None.
event_platform=Mac event_simpleName=ProcessRollup2 aid=<aid>
| join type=outer ContextProcessId_decimal
[search event_platform=Mac aid=<aid> event_simpleName=EndOfProcess
| rename _time as EndTime
| fields aid,ContextProcessId_decimal, EndTime]
| eval duration=if(isnull(EndTime),now()-_time,EndTime-_time)
| join type=outer aid,ProcessGroupId_decimal
[search event_platform=Mac event_simpleName=NetworkConnect* aid=<aid>
| stats count as NetworkConnectionCount by aid, ContextProcessId_decimal
| rename ContextProcessId_decimal as ProcessGroupId_decimal]
| search duration>86399 NetworkConnectionCount<5

(this image is based on an older version of the query, I can’t generate a new one
because the data has already rolled off)

Long-running process with few network connections - with filtering
Analyst notes
Because we’re using subsearches and subsearches have stricter memory constraints, time
frame is a concern for this query. The default Splunk configuration only allows 10,000 results to
be returned per subsearch.
Inventory management tools may also trigger this, and may need to be filtered out.

Known false positives
Calculating the process duration is based on the results returned by a subsearch. If the number
of EndOfProcess records is more than the limit on subsearches, the results will be incomplete.
That will cause the duration for some of the processes to be incorrectly calculated based on the
current time and artificially inflate the process duration.
Calculating the number of network connections is also based on the results of a subsearch. If
that subsearch returns incomplete results, the NetworkConnectionCount may be lower than
reality and contribute to a false positive.
If a process is terminated uncleanly, the EndOfProcess record may not be present. In these
cases, the process duration will be artificially inflated.

Known false negatives
None.
event_platform=Mac event_simpleName=ProcessRollup2 aid=<aid> NOT
[search event_platform=Mac event_simpleName=ProcessRollup2 (CommandLine=<yourC2>
OR ...) aid=<aid>
| fields ProcessGroupId_decimal]
| join type=outer TargetProcessId_decimal
[search event_platform=Mac aid=<aid> event_simpleName=EndOfProcess
| rename _time as EndTime
| fields aid,TargetProcessId_decimal, EndTime]
| eval duration=if(isnull(EndTime),now()-_time,EndTime-_time)
| join type=outer aid,ProcessGroupId_decimal
[search event_platform=Mac event_simpleName=NetworkConnect* aid=<aid>
| stats count as NetworkConnectionCount by aid, ContextProcessId_decimal
| rename ContextProcessId_decimal as ProcessGroupId_decimal]
| search duration>86399 NetworkConnectionCount<5

Process tree contains both sh and launchctl
Analyst notes
If a process tree contains both these commands, it means that one process ran a shell of some
sort and then launched an app. There are legitimate reasons for this to happen, but it is
relatively rare in our environment.

Known false positives
We have a process tree in our environment that triggers this, not sure of the exact cause but we
filter it out.

Known false negatives
Anything that matches the keywords filtered out at the end.
event_platform=Mac aid=<aid> event_simpleName=*ProcessRollup2 (sh OR launchctl)
| transaction aid,ProcessGroupId_decimal
| search sh launchctl NOT <redacted>

Process tree contains lots of shells like bash or sh
Analyst notes
Some filtering will likely be required. Experiment and adjust as needed.

Known false positives
Internal management tools may match this query.

Known false negatives
None.
event_platform=Mac event_simpleName=*ProcessRollup2 (CommandLine=sh* OR
CommandLine=/bin/sh* OR CommandLine=/bin/bash) aid=<aid>
| stats values(CommandLine) as Commands,count by aid,ProcessGroupId_decimal
| search NOT <yourC2>
| search count>20

Chown commands run on hidden dirs in user directories
Analyst notes
I found a few obvious exceptions common in our environment, but once you filter out the false
positives in your environment this query should be reliable.

Known false positives
Tools that create hidden user dirs for configuration will sometimes match this.

Known false negatives
Any record matching the exclusion list will be dropped.
event_simpleName=*ProcessRollup2 event_platform=Mac chown NOT <redacted> NOT
CommandLine=<redacted>
| regex CommandLine="/Users/[a-z]+/\..*"

All of the blackwidow entries above are malicious.

Chmod commands run on hidden dirs in user directories
Analyst notes
Very similar to the chown commands query.

Known false positives
In our environment, there are a bunch of developer tools that create hidden directories. The
known ones are filtered below.

Known false negatives
Any record matching the exclusion list will be dropped.
event_simpleName=*ProcessRollup2 event_platform=Mac chmod NOT <redacted>
| regex CommandLine="/Users/[a-z]+/\..*"
| table CommandLine

All of the blackwidow entries above are malicious.

Lots of profiling commands in the process tree
Analyst notes
Inventory management systems that run profiling commands will match this query. You will likely
need to do some filtering.

Known false positives
Internal tools that profile hardware.

Known false negatives
Any record matching the exclusion list will be dropped.
event_platform=Mac event_simpleName=ProcessRollup2 aid=<aid> (networksetup OR who OR
whoami OR sysctl)
| eval JoinId=ParentProcessId_decimal
| rename CommandLine as ChildCommandLine
| join type=outer aid,JoinId
[search event_platform=Mac aid=0000de4aa30a4a616d0bb3bf5986eadc
event_simpleName=ProcessRollup2
| eval JoinId=TargetProcessId_decimal
| rename CommandLine as ParentCommandLine]
| search NOT ChildCommandLine=<redacted>
| search NOT ParentCommandLine=<redacted>
| stats values(ChildCommandLine) as Commands, count by aid
| search count>1

Rare processes with specific characteristics
Rare hashes using security_authtrampoline unfiltered (so you can play with
thresholds)
Analyst notes
When a process wants upgraded credentials, it can use security_authtrampoline. Legitimate
installers will do this as well. Due to limitations with the Splunk configuration, I couldn’t use
lookup tables to get a full count of how common each hash is. That’s why this query uses
top/rare commands to filter on hashes that appear on more than one host.
This particular query will give more raw results for people who want to play with their own
thresholds.

Known false positives
Legitimate installers that are rare.
The filter for rare/common hashes depends on a subsearch which has limits on the number of
records returned. Hashes that are common but not common enough may not be filtered out.
Likewise, hashes that are rare but not rare enough may not be filtering out.

Known false negatives
None.

event_platform=Mac event_simpleName=*ProcessRollup2
[search event_platform=Mac event_simpleName=*ProcessRollup2
security_authtrampoline
| fields ProcessGroupId_decimal]
| dedup aid, SHA256HashData
| eval CommandLine=substr(CommandLine,1,100)
| stats values(CommandLine) as Commands, dc(aid) as AuthtrampolineCount by
SHA256HashData
| eventstats sum(AuthtrampolineCount) as AuthtrampolineTotal
| eval AuthTrampolinePerc=round((AuthtrampolineCount/AuthtrampolineTotal)*100,7)
| join type=outer SHA256HashData
[search event_platform=Mac event_simpleName=*ProcessRollup2
| rare SHA256HashData limit=10000 by aid
| stats dc(aid) as RareGPopCount by SHA256HashData
| eventstats sum(RareGPopCount) as RareGPopTotal
| eval RareGPopPerc=round((RareGPopCount/RareGPopTotal)*100,7) ]
| join type=outer SHA256HashData
[search event_platform=Mac event_simpleName=*ProcessRollup2
| top SHA256HashData limit=10000 by aid
| stats dc(aid) as CommonGPopCount by SHA256HashData
| eventstats sum(CommonGPopCount) as CommonGPopTotal
| eval CommonGPopPerc=round((CommonGPopCount/CommonGPopTotal)*100,7)]
| fillnull value=0 CommonGPopCount
| fillnull value=0 RareGPopCount

Rare hashes using security_authtrampoline
Analyst notes
When a process wants upgraded credentials, it can use security_authtrampoline. Legitimate
installers will do this as well. Due to limitations with the Splunk configuration, I couldn’t use
lookup tables to get a full count of how common each hash is. That’s why this query uses
top/rare commands to filter on hashes that appear on more than one host.
An additional level of filtering is required. Once you have the list of suspicious hashes, you
should count the instances of the suspicious hashes per agent. The screenshots below show
this.

Known false positives
Legitimate installers that are rare.

The filter for rare/common hashes depends on a subsearch which has limits on the number of
records returned. Hashes that are common but not common enough may not be filtered out.
Likewise, hashes that are rare but not rare enough may not be filtering out.

Known false negatives
If a process is not rare enough, it will be filtered out. If you suspect that the malware you’re
hunting for already has significant penetration, you will need to filter based on counts in the
unaffected population.
event_platform=Mac event_simpleName=*ProcessRollup2
[search event_platform=Mac event_simpleName=*ProcessRollup2
security_authtrampoline
| fields ProcessGroupId_decimal]
| dedup aid, SHA256HashData
| eval CommandLine=substr(CommandLine,1,100)
| stats values(CommandLine) as Commands, dc(aid) as AuthtrampolineCount by
SHA256HashData
| search AuthtrampolineCount=1
| join type=outer SHA256HashData
[search event_platform=Mac event_simpleName=*ProcessRollup2
| rare SHA256HashData limit=10000 by aid
| stats dc(aid) as RareGPopCount by SHA256HashData]
| join type=outer SHA256HashData
[search event_platform=Mac event_simpleName=*ProcessRollup2
| top SHA256HashData limit=10000 by aid
| stats dc(aid) as CommonGPopCount by SHA256HashData]
| fillnull value=0 CommonGPopCount
| fillnull value=0 RareGPopCount
| search CommonGPopCount<2 RareGPopCount<2
| eval Auth_Common_Rare=AuthtrampolineCount.",".CommonGPopCount.",".RareGPopCount
| fields SHA256HashData, Commands, Auth_Common_Rare

Now looking for how many agents reported these hashes . . .

Self-deleting rare processes
Analyst notes
Some installers will self-delete, so you have to filter on rare/common processes.

Known false positives
The filter for rare/common hashes depends on a subsearch which has limits on the number of
records returned. Hashes that are common but not common enough may not be filtered out.
Likewise, hashes that are rare but not rare enough may not be filtered out.

Known false negatives
None.
event_platform=Mac event_simpleName=ProcessSelfDeleted
| map search="search event_simpleName=*ProcessRollup2 aid=$aid$
TargetProcessId_decimal=$ContextProcessId_decimal$"
| dedup aid,SHA256HashData
| eval CommandLine=substr(CommandLine,1,50)
| stats values(CommandLine) as Commands, dc(aid) as UniqueAgentCount by
SHA256HashData
| join type=outer SHA256HashData
[search event_platform=Mac event_simpleName=*ProcessRollup2
| top SHA256HashData limit=10000 by aid
| stats dc(aid) as CommonGPopCount by SHA256HashData]
| join type=outer SHA256HashData
[search event_platform=Mac event_simpleName=*ProcessRollup2
| rare SHA256HashData limit=10000 by aid
| stats dc(aid) as RareGPopCount by SHA256HashData]
| fillnull value=0 CommonGPopCount
| fillnull value=0 RareGPopCount

| search UniqueAgentCount=1 CommonGPopCount<2 RareGPopCount<2

Rare launch agents
Analyst notes
This query will be more reliable run over smaller timeframes.

Known false positives
None.

Known false negatives
None.
event_platform=Mac event_simpleName=*ProcessRollup2 CommandLine=*LaunchAgents*
| dedup aid,CommandLine | makemv CommandLine delim=" "
| eval CommandLine=mvfilter(match(CommandLine, ".*LaunchAgents.*"))
| eval CommandLine=replace(CommandLine,"/Users/[a-z]+/", "/") | eval
CommandLine=replace(CommandLine,"\"$", "")
| dedup aid,CommandLine
| stats count by CommandLine
| sort count

Informational
Count network connection counts for a process
Analyst notes
This query depends on a subsearch and will run into subsearch memory limits. In rare cases or
over long timeframes, the count may be lower than reality.

Known false positives
None.

Known false negatives
None.
event_platform=Mac event_simpleName=*ProcessRollup2 <ProcessGroupId_decimal>
aid=<aid>
| join aid,ProcessGroupId_decimal
[search event_platform=Mac <ProcessGroupId_decimal> aid=<aid>
event_simpleName=NetworkConnect*
| stats count as NetworkConnectionCount by aid, ContextProcessId_decimal
| rename ContextProcessId_decimal as ProcessGroupId_decimal]
| stats count by aid,ProcessGroupId_decimal

